**********************************************

*            State machine                          *

**********************************************

$nolist

*uC 908QT2 REGISTERS

prtA    equ  $00 ;portA

ddrA    equ  $04 ;direction prtA register

config2 equ  $1E

config1 equ  $1F

TSC     equ  $20 ;Timer Status & Control reg.

TSTOP   equ  5   ;Timer Stop bit

TmodH   equ  $23 ;Timer modulo reg, High

TmodL   equ  $24 ;Timer modulo reg, Low

ADSCR   equ  $3C ;A/D Status &  Control reg.

COCO    equ  7   ;A/D conversion complete flag

ADR     equ  $3E ;A/D Data register

ADIclk  equ  $3F ;A/D input clock register

RAM     equ  $0080

ROM     equ  $F800

$list

**********************************************

*I/O

LED0    equ 0 ;prtA,output

LED1    equ 1 ;prtA,output

lamp    equ 4 ;prtA,output

phCell  equ 5 ;prtA,input ADC

*EQUATES

Vp      equ $52   ;>1.6V night photocell threshold

dark    equ 0 ;flag for dark

*VARIABLES

          org   RAM

NT        rmb   1 ;delay register

flag      rmb   1 ;flag register

cntSEC    rmb   1 ;seconds counter

*INITIALIZATION

          org   ROM

****************************************************

*    DATA STRUCTURE

out       equ   0 ;offset for prtA

delay     equ   1 ;offset for delay

next0     equ   2 ;offset for next state when dark=0

next1     equ   3 ;offset for next state when dark=1

states    equ   * ;base address for state

initST    equ   *-states ;offset for initial state

*****************************************************

State1    equ   *-states  ;offset for state1

          fcb   %00000001 ;prtA for state1

          fcb   2T        ;delay for state1=2sec

          fcb   state1    ;next state for dark=0

          fcb   state2    ;next state for dark=1

State2    equ   *-states

          fcb   %00010010

          fcb   240T      ;delay for state2=4min

          fcb   state3    ;next state for dark=0

          fcb   state3    ;next state for dark=1

State3    equ   *-states

          fcb   %00000011

          fcb   2T        ;delay for state3=2sec

          fcb   state1    ;next state for dark=0

          fcb   state3    ;next state for dark=1

*****************************************************

init      bset  0,config1 ;COP disable

          clr   config2   ;set internal Osc,

          lda   $ffc0 ;set fosc=12.8 MHz

          sta   $0038

          clr   NT

          clr   flag

          clr   cntSEC

          clr   prtA

          mov   #%00010011,ddrA  ;set I/O prtA

          mov   #%01110110,TSC   ;TOFint.enable,

                ;Stop & clear Timer, set PRESCALER=64

          mov   #%01000000,ADIclk;set fADC=fBUS/4=.8MHz

          mov   #$c3,TmodH       ;set TOF period=1sec

          mov   #$50,TmodL       ;    by $c350-->TMOD

          bclr  TSTOP,TSC        ;start TIMER

          ldx   #initST          ;state1-->X

          clrH

          cli                    ;set Interrupt enable

.page

Main      lda   states+out,x   ;set current State prtA

          sta   prtA

          lda   states+delay,x ;set delay of curr.State

          sta   NT

          jsr   dly            ;execute delay

nextST    jsr   photoCHK       ;What is the next state?

          brclr dark,flag,dark0

dark1     ldx   states+next1,x

          bra   Main

dark0     ldx   states+next0,x

          bra   Main

********************************************************

dly       lda   cntSEC

          add   NT

dN        cmp   cntSEC

          bne   dN

          rts

********************************************************

PhotoChk  bclr  dark,flag       ;0-->dark flag

          mov   #%00000011,ADscr;start ADC3 pA5

          brclr COCO,ADscr,*    ;wait for ADC3 end

          lda   ADR             ;ADC3 data-->Acc

          cmp   #Vp             ;check for a Night

          bls   phEND           ;ADC3 data<=Vp?

          bset  dark,flag       ;1-->dark flag

phEND     rts

********************************************************

TimInt    inc   cntSEC  ;increment cntSEC every TOF=1sec

          lda   TSC     ;reset TOF

          bclr  7,TSC   ;0-->TOF bit

          rti

***************************************

          org    $fff2 ;TOF Vector

          fdb    TimInt

          org    $fffe ;Reset Vector

          fdb    init

.end

$nolist

